Mathematica code to solve a certain degree 10 Diophantine equation in 3 variables under some conditions - Ziqing Xiang

This is the Mathematica code for the joint work with Eiichi Bannai, Etsuko Bannai, Wei-Hsuan Yu and Yan Zhu entitled "Classification of spherical 2-distance {4, 2, 1}-design".

Section 3

Definition of Gegenbauer polynomial $Q_{n,4}(\xi)$.

```
In[1]:= Q[n_, 4, ξ_] := n (n + 6) / 24 ((n^2 + 6 n + 8) ξ^4 - 6 (n + 2) ξ^2 + 3);
```

Definition of F. Eq. (3.3).

```
In[2]:= F = Factor[Q[n, 4, 1] + k Q[n, 4, y / k] + (v - k - 1) Q[n, 4, (-y - 1) / (v - k - 1)] /.
        v → 1 / μ (k - x) (k - y) /. k → μ - x y /. n → (μ - x y) (μ - x y - x) (x + 1) / μ / (x - y)]
```

Out[2]=

$$\Big((x + x y - \mu)\ (y + x y - \mu)^2 \left(x^2 y + x^3 y + x^2 y^2 + x^3 y^2 + 5 x \mu - x^2 \mu - 6 y \mu - 2 x y \mu - 2 x^2 y \mu + \mu^2 + x \mu^2\right) \left(x^2 y + x^3 y + x^2 y^2 + x^3 y^2 + 3 x \mu - x^2 \mu - 4 y \mu - 2 x y \mu - 2 x^2 y \mu + \mu^2 + x \mu^2\right) \left(x^5 y^2 + 2 x^6 y^2 + x^7 y^2 - x^2 y^3 - 2 x^3 y^3 - x^4 y^3 + x^5 y^3 + 2 x^6 y^3 + x^7 y^3 - x^2 y^4 - 2 x^3 y^4 - x^4 y^4 - 5 x^4 y \mu - 7 x^5 y \mu - 2 x^6 y \mu - x y^2 \mu - 2 x^2 y^2 \mu + 3 x^3 y^2 \mu - x^4 y^2 \mu - 8 x^5 y^2 \mu - 3 x^6 y^2 \mu + 2 y^3 \mu + 4 x y^3 \mu - x^2 y^3 \mu - 3 x^3 y^3 \mu + 6 x^3 \mu^2 + 5 x^4 \mu^2 + x^5 \mu^2 - 2 x y \mu^2 - 7 x^2 y \mu^2 + 6 x^3 y \mu^2 + 10 x^4 y \mu^2 + 3 x^5 y \mu^2 + 3 y^2 \mu^2 + 8 x y^2 \mu^2 + 3 x^2 y^2 \mu^2 - 3 x \mu^3 - 6 x^2 \mu^3 - 4 x^3 \mu^3 - x^4 \mu^3 + y \mu^3 + x y \mu^3\right)\Big) \Big/ \left(24\ (1 + x)^2\ (x - y)^4\ (x y - \mu)^2\ \mu^4\right)$$

The four factors F_0, F_1, F_2, F_3 of F. Eqs. (3.4) - (3.7).

```
In[3]:= F0 = (μ - x - x y) (μ - y - x y)^2 / (24 μ^4 (1 + x)^2 (x - y)^4 (μ - x y)^2);
F1 = μ^2 + 5 μ x + μ^2 x - μ x^2 - 6 μ y - 2 μ x y + x^2 y - 2 μ x^2 y + x^3 y + x^2 y^2 + x^3 y^2 ;
F2 = μ^2 + 3 μ x + μ^2 x - μ x^2 - 4 μ y - 2 μ x y + x^2 y - 2 μ x^2 y + x^3 y + x^2 y^2 + x^3 y^2;
F3 = 3 μ^3 x + 6 μ^3 x^2 - 6 μ^2 x^3 + 4 μ^3 x^3 - 5 μ^2 x^4 + μ^3 x^4 - μ^2 x^5 - μ^3 y + 2 μ^2 x y - μ^3 x y + 7 μ^2 x^2 y -
    6 μ^2 x^3 y + 5 μ x^4 y - 10 μ^2 x^4 y + 7 μ x^5 y - 3 μ^2 x^5 y + 2 μ x^6 y - 3 μ^2 y^2 + μ x y^2 - 8 μ^2 x y^2 +
    2 μ x^2 y^2 - 3 μ^2 x^2 y^2 - 3 μ x^3 y^2 + μ x^4 y^2 - x^5 y^2 + 8 μ x^5 y^2 - 2 x^6 y^2 + 3 μ x^6 y^2 - x^7 y^2 - 2 μ y^3 -
    4 μ x y^3 + x^2 y^3 + μ x^2 y^3 + 2 x^3 y^3 + 3 μ x^3 y^3 + x^4 y^3 - x^5 y^3 - 2 x^6 y^3 - x^7 y^3 + x^2 y^4 + 2 x^3 y^4 + x^4 y^4;
```

Verify the factorization.

```
In[7]:= Simplify[F0 F1 F2 F3 - F]
```

Out[7]= 0

Section 5.2

Alternative definition of n and v. Eqs. (5.1) and (5.2).

In[8]:=
```
n = (x + 1) (μ - x y) (μ - x y - x) / (μ (-y + x));
v = n + 1 - (y n + μ - x y) / x;
```

Step 1

Assumption on y in Step 1. Eq. (5.3).

In[10]:=
```
yassum = y ≤ - (2 x^3 + 3 x^2 + 3 x + 2) || - (2 x^3 + 3 x^2 - 3 x - 3) ≤ y ≤ -1;
```

The computer proof for Step 1.

In[11]:=
```
Simplify[v > n (n + 3) / 2 || F3 > 0, x ≥ 1 && μ ≥ 1 && yassum]
```

Out[11]=
```
True
```

Step 2

Definition of a. Eq. (5.4).

In[12]:=
```
usub = μ → - (x + a) y;
```

Definition of G_1.

In[13]:=
```
G1 = Factor[(F3 /. usub) / y^2];
```

The computer proof for Step 2(a).

In[14]:=
```
Simplify[(G1 /. y → 0) < 0, 2 ≤ x && -x < a]
```

Out[14]=
```
True
```

The computer proof for Step 2(b).

In[15]:=
```
Simplify[(G1 /. y → -1) > 0, 2 ≤ x && -x < a]
```

Out[15]=
```
True
```

The computer proof for Step 2(c).

In[16]:=
```
Simplify[(G1 /. y → - (2 x^3 + 3 x^2 + 3 x + 2)) > 0, 2 ≤ x && (-x < a ≤ -1 || 3 ≤ a)]
```

Out[16]=
```
True
```

Step 3

Definition of b. Eq. (5.11).

In[17]:=

$$\text{ysub} = y \to -\left(2 x^3 + 3 x^2 + \frac{3}{2} (-1 + a)\, a\, x - \frac{3}{2} (-1 + a)^2 a + \frac{1}{4 x} 3 (-1 + a)\, a \left(2 - 4 a + 3 a^2\right) - \frac{1}{4 x^2} 3 (-1 + a)\, a^2 \left(3 - 6 a + 4 a^2\right) + \frac{1}{8 x^3} 3 (-1 + a)\, a \left(5 - 9 a + 16 a^2 - 20 a^3 + 11 a^4\right) + \frac{b}{x^4}\right);$$

Definition of G_2.

In[18]:=
```
G2 = Factor[G1 /. ysub];
```

The computer proof for Step 3(a).

In[19]:=
```
Simplify[(G2 /. b → -3994) > 0, 90 ≤ x && -1 ≤ a ≤ 3]
```

Out[19]=
```
True
```

The computer proof for Step 3(b).

In[20]:=
```
Simplify[(G2 /. b → 64) < 0, 90 ≤ x && -1 ≤ a ≤ 3]
```

Out[20]=
```
True
```

Step 4

Definition of m^2. Eq. (5.12).

In[21]:=
```
mmsub = mm → n - (4 x^2 + 4 x - 2);
```

Definition of G_3.

In[22]:=
```
G3 = Factor[mm /. mmsub /. usub /. ysub];
```

Definition of $\tilde{m}^2$. Eq. (5.12).

In[23]:=

$$\text{mmtsub} = \text{mmt} \to a^2 - \frac{(-1+a)\,a^2}{x} + \frac{(-1+a)\,a\,\left(1+a^2\right)}{x^2} - \frac{1}{2\,x^3}\,(-1+a)\,a\,\left(1+2\,a+2\,a^3\right) + \frac{1}{4\,x^4}\,(-1+a)\,a\,\left(7-a+4\,a^2+4\,a^4\right) + \frac{c}{x^5};$$

Definition of G_4.

In[24]:=
```
G4 = Factor[mmt /. mmtsub];
```

The computer proof for Step 4(a).

In[25]:=
```
Simplify[G3 > (G4 /. c → -1620), 90 ≤ x && -1 ≤ a ≤ 3 && -3994 ≤ b ≤ 64]
```

Out[25]=
```
True
```

The computer proof for Step 4(b).

In[26]:=
```
Simplify[G3 < (G4 /. c → 3), 90 ≤ x && -1 ≤ a ≤ 3 && -3994 ≤ b ≤ 64]
```

Out[26]=
```
True
```

The computer proof for Step 4(c).

In[27]:=
```
Simplify[G4 < 9, 90 ≤ x && -1 ≤ a ≤ 3 && -1620 ≤ c ≤ 3]
```

Out[27]=
```
True
```

Step 5

Definition of $\tilde{n}$. Eq. (5.16).

In[28]:=
```
nt = mmt + (4 x^2 + 4 x - 2);
```

Definition of $\tilde{v}$. Eq. (5.16).

In[29]:=
```
vt = nt + 1 - (y nt + μ - x y) / x;
```

Definition of $\tilde{z}$. Eq. (5.15).

In[30]:=
```
zt = 144 mmt - (3 vt + (8 y + 4 x^3 + 6 x^2 + 3) (2 x + 1) - 3 / 2 mmt (mmt - 7)) ^2;
```

Definition of G_5.

In[31]:=
```
G5 = Factor[zt /. usub /. ysub /. mmtsub];
```

The computer proof for Step 5(a). We use maximum value as an upper bound when $-5 \le i \le -2$, and use the maximum absolute sum as an upper bound when $-20 \le i \le -6$.

In[32]:=
```
coeff = Table[If[i ≥ -5,
    Maximize[{Abs[Coefficient[G5, x^i]],
       -1 ≤ a ≤ 3 && -3994 ≤ b ≤ 64 && -1620 ≤ c ≤ 3}, {x, a, b, c}][[1]],
    FromCoefficientRules[Map[Abs, CoefficientRules[Coefficient[G5, x^i],
          {x, a, b, c}], {2}], {x, a, b, c}] /. a → 3 /. b → 3994 /. c → 1620
   ],
   {i,
    -20,
    -2}]
```

Out[32]= $\left\{545102954553600, 1801659993602400, \frac{5886126860798565}{2}, 3236151695060880, \frac{19524150631163025}{8}, \frac{23280405487750863}{16}, 741299023799055, 334643780679111, 131234047150977, 44282400488163, 13051286303076, 3426985781691, \frac{26026504017233}{32}, \frac{639099553189}{4}, \frac{37583702399}{2}, 17107740, 3629457, 1222632, 3672\right\}$

The computer proof for Step 5(b).

In[33]:=
```
Simplify[Total[coeff x^Range[-20, -2]] < 1, x ≥ 120]
```

Out[33]=
```
True
```

Step 6

Definition of G_6.

In[34]:=
```
G6 = Factor[μ (y - x) (mm - m0^2) /. mmsub];
```

Definition of F_4. Eq. (5.18)

In[35]:=
```
F4 =
  Collect[Factor[Denominator[Factor[PolynomialExtendedGCD[F3, G6, μ]][[2]]][[1]] /
     (x^2 (1 + x)^2 (x - y) y^2 (1 + y))], y, Factor]
```

Out[35]=

$$
\begin{aligned}
&x^5 \left(-4 + m0^2 + 4 x + 4 x^2\right) \left(-3 + m0^2 + 4 x + 4 x^2\right) \\
&\quad \left(-3 + 3 m0^2 + 6 x + 3 m0^2 x + 21 x^2 + m0^2 x^2 + 16 x^3 + 4 x^4\right) - \\
&x^2 \left(3 m0^2 - 6 m0^4 + 3 m0^6 + 12 x - 51 m0^2 x + 36 m0^4 x + 3 m0^6 x - 204 x^2 + 204 m0^2 x^2 + \right. \\
&\quad 37 m0^4 x^2 + 5 m0^6 x^2 + 204 x^3 + 305 m0^2 x^3 + 60 m0^4 x^3 + 3 m0^6 x^3 + 1048 x^4 + \\
&\quad 179 m0^2 x^4 + 80 m0^4 x^4 + m0^6 x^4 + 688 x^5 + 352 m0^2 x^5 + 48 m0^4 x^5 + 252 x^6 + \\
&\quad \left. 448 m0^2 x^6 + 12 m0^4 x^6 + 704 x^7 + 240 m0^2 x^7 + 832 x^8 + 48 m0^2 x^8 + 384 x^9 + 64 x^{10}\right) y + \\
&x \left(12 m0^2 - 16 m0^4 + 4 m0^6 + 48 x - 110 m0^2 x + 21 m0^4 x + 3 m0^6 x - 244 x^2 - \right. \\
&\quad 56 m0^2 x^2 + 34 m0^4 x^2 + 2 m0^6 x^2 - 448 x^3 + 38 m0^2 x^3 + 30 m0^4 x^3 - \\
&\quad \left. 180 x^4 + 8 m0^2 x^4 + 12 m0^4 x^4 - 248 x^5 - 448 x^6 - 288 x^7 - 64 x^8\right) y^2 + \\
&\left(-12 m0^2 + 7 m0^4 - m0^6 - 48 x + 20 m0^2 x - 44 x^2 + 32 m0^2 x^2 - 8 x^3 + \right. \\
&\quad \left. 24 m0^2 x^3 - 44 x^4 + 12 m0^2 x^4 - 48 x^5 - 16 x^6\right) y^3
\end{aligned}
$$

Step 7

Definition of $y^{(1)}$, $y^{(2)}$, $y^{(3)}$.

In[36]:=
```
y1 = -(2 x^3 + 3 x^2 + 3/2 m0 (m0 + 1) x + 3/4 m0 (m0 + 1));
y2 = -(2 x^3 + 3 x^2 + 3/2 m0 (m0 - 1) x + 3/4 m0 (m0 - 1));
y3 = x;
```

The computer proof for Step 7(a).

In[39]:=
```
Simplify[(F4 /. y → y1) == 0, m0 == 0]
```

Out[39]=
```
True
```

The computer proof for Step 7(b).

In[40]:=
```
Simplify[(F4 /. y → y1 - 1 / 2) > 0, x ≥ 90 && 1 ≤ m0 ≤ 2]
```

Out[40]=
```
True
```

The computer proof for Step 7(c).

In[41]:=
```
Simplify[(F4 /. y → y1 + 1 / 2) < 0, x ≥ 90 && 1 ≤ m0 ≤ 2]
```

Out[41]=
```
True
```

The computer proof for Step 7(d).

In[42]:=
```
Simplify[(F4 /. y → y2 + 1 / x) < 0, x ≥ 90 && m0 == 0]
```

Out[42]=
```
True
```

The computer proof for Step 7(e).

In[43]:=
```
Simplify[(F4 /. y → y2 - 1 / 2) < 0, x ≥ 90 && 1 ≤ m0 ≤ 2]
```

Out[43]=
```
True
```

The computer proof for Step 7(f).

In[44]:=
```
Simplify[(F4 /. y → y2 + 1 / 2) > 0, x ≥ 90 && 0 ≤ m0 ≤ 2]
```

Out[44]=
```
True
```

The computer proof for Step 7(g).

In[45]:=
```
Simplify[(F4 /. y → y3 - 1) > 0, x ≥ 1 && 0 ≤ m0 ≤ 2]
```

Out[45]=
```
True
```

The computer proof for Step 7(h).

In[46]:=
```
Simplify[(F4 /. y → y3 + 1) < 0, x ≥ 1 && 0 ≤ m0 ≤ 2]
```

Out[46]=
```
True
```

Step 8

Speed up Step 8 using multiple kernels.

In[47]:=
```
LaunchKernels[8];
```

The computer proof for Step 8.

In[48]:=
```
(* Step 8a. The computer proof *)
Select[DeleteDuplicates[Flatten[ParallelTable[
    If[(μ /. #) > 0 && y0 ≤ -1 && y0 ≠ -(2 x0^3 + 3 x0^2), {x0, y0, μ /. #}, {}] & /@
     Solve[(F3 /. x → x0 /. y → y0) == 0, μ, Integers]
    , {x0, 1, 120}, {y0, -(2 x0^3 + 3 x0^2 + 3 x0 + 2), -(2 x0^3 + 3 x0^2 - 3 x0 - 3)}], 2]],
 Length[#] > 0 &]
```

Out[48]=
```
{{1, -1, 1}}
```